/* прим. пер.: все примеры в этой книге при создании и сохранении модулей используют нотацию типа C:Start. Начиная с версии CA-Visual Objects 2.5b (см. файл Watsnew.doc), в именах модулей более не допускается использование символов \ / : * ? " < > | .*/

К вопросу о раздвоении личности


Говорят, что хороший психиатр может многое рассказать о личности художника по его работе, впрочем, это касается не только художников. Дизайн интерфейса и выбор данных для тестирования могут открыть ваши самые сокровенные секреты опытному взору.

Слава Богу, не похоже, чтобы многие психиатры читали эту книгу, а поэтому мы можем сэкономить и продолжим использовать тот же набор видеофильмов, что и ранее. Тем не менее, психоанализ лица, разработавшего интерфейс Проводника ОС Windows95, мог бы оказаться весьма интересным. Окно разделено на две части перемещаемой вертикальной чертой; в одной части - древовидная структура, в другой - списочная. Если его разрабатывал один человек, то не может ли это указывать на раздвоение его личности? Или это была работа команды, так и не пришедшей к согласию? Как бы то ни было, мы получили три элемента управления - разделенное окно в виде SplitWindow, древовидную структуру в виде TreeView и список в виде ListView.

В этой главе мы используем все три элемента, добавим элемент управления Вкладки - Tabs и создадим приложение, полностью отличающееся от всего созданного нами ранее. Для этого приложения нет файлов данных, мы создадим их по мере продвижения, а посему приступим.

В каталоге C:\EVO создайте подкаталог F1 и скопируйте в него все файлы из каталога Ch21 сопроводительного диска CD. Далее, запустите CA-Visual Objects, выберите проект EVO. В меню выберите создание нового приложения, перейдите на вкладку Basic, там выберите Empty Application, отключите флажок ReadMe и включите в состав приложения библиотеки System Library, GUI Library и RDD Library. В завершение измените название приложения на Ch21 и нажмите кнопку ОК.

Разделенное окно - SplitWindows
Во вновь созданном приложении измените наименования модуля, созданного по умолчанию, на C:Start. Откройте редактор кода и введите следующее:

METHOD Start( ) CLASS App

LOCAL oShell AS ShellWindow

LOCAL oSplit AS SplitWindow

oShell := ShellWindow{SELF}

oShell:Caption := 'Формула 1'

oShell:Size := Dimension{510, 350}

oShell:EnableStatusBar(FALSE)

oShell:Show(SHOWCENTERED)

oSplit := SplitWindow{oShell, TRUE, TRUE}

oSplit:Size := Dimension{500, 300}

oSplit:Layout := Dimension{7, 2}

oSplit:Show(SHOWZOOMED)

SELF:Exec( )

RETURN NIL
 Сохраните код, закройте редактор, скомпилируйте и запустите приложение. В соответствии с приведенным выше кодом мы создаем окно оболочки приложения  - ShellWindow, присваиваем значение заголовку этого окна, указываем его размер, после чего центрируем его на экране. Все это мы видели и ранее, однако дальше следует новый для нас код.

[image: image1.png]Puc.21.1

SplitWindow ¢ AByMA pAaaMI
OKOH, 10 CeMb OKOH B KaKAOM.





В следующей строке мы создаем окно SplitWindow и  передаем ему три параметра. Первый - <oShell>, указывает на то, что владельцем данного окно является окно оболочки, т.е. ShellWindow. Второй параметр - TRUE указывает на то, что разрешено перемещение горизонтальных разделителей, а третий, также TRUE, означает то же самое для разделителей вертикальных. Вскоре мы увидим, что это означает на практике, а пока продолжим изучение кода. В следующей строке устанавливается размер окна, а затем SplitWindow:Layout ACCESS/ASSIGN устанавливается в значение Dimension 7, 2. Как мы видим из рис. 21.1, это означает, что мы получаем два ряда окон, по семь в каждом из них, что дает нам 14 окон, в которым мы можем выводить нужные нам данные. После этого окно максимально разворачивается и запускается цепочка Exec, где в данный момент мы и находимся.

Сейчас можно продемонстрировать эффект второго и третьего параметров, переданных SplitWindow. Поместите курсор мыши над одним из горизонтальных разделителей окон. При изменении формы курсора нажмите левую кнопку мыши и переместите разделитель вверх или вниз. Как видно из рисунка 21.2, перемещается только одна секция разделителя между вертикальными границами окна.

[image: image2.png]Puc.21.2

MepemeLieHne ropu3oHTanHoOro
pasgenuTens & okte SplitWindow.
MepeMmeLLaeTcs TONLKO €ro YacTb,
pacnonoxeHHas Mexay
BEPTUKANIbHBIMM PA3AEnUTENAMA.

i Formula 1

=lolx|





Из рис. 21.3 видно, что в отличие от горизонтального, вертикальный разделитель перемещается полностью.

Теперь поместите курсор на пересечение горизонтального и вертикального разделителей. Когда курсор изменит форму на крест, вы сможете одновременно переместить и вертикальный, и горизонтальный разделители (рис. 21.4). Последнее, что нам осталось, это попытаться изменить размеры самого окна оболочки. Обратите внимание, что размеры внутренних окон при этом изменяются пропорционально. На этом можно завершить тестирование и закрыть приложение.

Связывание окна и панели

То, что мы сделали, является хорошей демонстрацией возможностей окна SplitWindow, однако этот пример не очень практичен. Вновь откройте метод Start() и измените его следующим образом.

METHOD Start( ) CLASS App

LOCAL oShell AS ShellWindow

LOCAL oSplit AS SplitWindow

LOCAL oShowWin AS ShowWin 

// Новая строка
oShell := ShellWindow{SELF}

oShell:Caption := "Формула 1"

oShell:Size := Dimension{510, 350}

oShell:Show(SHOWCENTERED)

oSplit := SplitWindow{oShell, TRUE, TRUE}

oSplit:Size := Dimension{500, 300}

oSplit:Layout := Dimension{2, 2}

// Изменено
oSplit:Show(SHOWZOOMED)

oShowWin := ShowWin{oSplit} 

// Новая строка
oSplit:SetPaneSize(oShowWin:Size, 1) 
// Новая строка
oShowWin:Show( ) 


// Новая строка

oSplit:SetPaneClient(oShowWin, 1) 
// Новая строка
SELF:Exec( )

RETURN NIL
Как вы можете видеть, у нас появилась новая переменная для класса ShowWin, а параметры для oSplit:Layout изменились на {2, 2}, что в итоге дает нам четыре одинаковых окна.

Новый код начинается с создания переменной для класса ShowWin, а SplitWindows передается классу ShowWin в качестве владельца (мы создадим класс ShowWin как только разберем наш код). В следующей строке - oSplit:SetPaneSize, мы устанавливаем размеры нового объекта <oShowWin> и соответствующий размер первой панели. Далее следует вызов метода Show() и устанавливаются отношения между <oShowWin> и <oSplit> в панели 1 для четырех окон, которые являются клиентами окна SplitWindow. Хотя это и выглядит довольно запутанным, все прояснится очень скоро, как только мы запустим следующий тест, но пока нам нужно создать класс ShowWin.

Создайте новый модуль, назовите его "W:ShowWin", откройте оконный редактор, выберите тип окна DIALOGWINDOW, назовите его "ShowWin" и нажмите кнопку ОК. Далее перейдите на вкладку DlgWindow и установите свойства окна Type в TabPage, Modeless - в Yes, а размеры - 155 х 84 (ширина и высота).  Затем перейдите на вкладку Styles, отключите заголовок окна, установив свойство Caption в значение No Caption Bar и сохраните окно.

Теперь нам нужны три элемента управления RadioButton со следующими свойствами:

Caption
Name 
Width 
Height 
Left 
Top

Drivers 
rbDrivers 
120 

18 

21 
13

Teams 
rbTeams 
120 

18 

21 
33

Circuits 
rbCircuits 
120 

18 

21 
53
Вновь сохраните окно, вернитесь на вкладку DlgWindow, откройте свойство PostInitAction и дополните код следующим:

METHOD PostInit(oParent,uExtra) CLASS ShowWin

//Put your PostInit additions here

SELF:oCCrbDrivers:Pressed := TRUE

RETURN NIL
В настоящий момент все, что может этот код - это сделать кнопку Drivers (пилоты) кнопкой по умолчанию. Сохраните код, закройте редактор, скомпилируйте и запустите приложение.

Как видно из рисунка 21.5, новое окно помещено в верхнюю левую панель, таким образом, эта панель является первой. Ее размер изменен в соответствии с размерами окна. Попробуйте переместить разделители и вы убедитесь, что они остались перемещаемыми, однако в окне не появляются автоматически строки прокруток, как это обычно происходит при достижении края окна. Также заметьте, что кнопка Drivers имеет фокус и все три кнопки работают как группа, хотя такой элемент как RadioButtonGroupBox в окне отсутствует. По окончании тестирования закройте приложение и давайте добавим еще немного кода.

Создание файлов данных и серверов

Далее нам нужны файлы данных, поэтому возвратитесь в модуль C:Start, откройте метод App:Start() и добавьте следующие строки:

METHOD Start( ) CLASS App

LOCAL oShell AS ShellWindow

LOCAL oSplit AS SplitWindow

LOCAL oShowWin AS ShowWin
SetDefault("C:\Evo\F1") 

// Новая строка

SetExclusive(FALSE) 

// Новая строка
IF ! File("Drivers.Dbf") 

// Новая строка
SELF:CreateDataFiles( ) 
// Новая строка
ENDIF 



// Новая строка

oShell := ShellWindow{SELF}

oShell:Caption := "Формула 1"
Сохраните код и взгляните на новые строки. Мы уже знакомы с функциями SetDefault и SetExclusive, однако следующая строка для нас нова. В этой строке мы используем функцию File() для того, чтобы определить наличие или отсутствие конкретного файла. Если файл отсутствует, осуществляется вызов метода CreateDataFiles() класса Арр. Как обычно, в настоящее время у нас нет такого метода. Для экономии времени импортируйте этот код.

Закройте редактор, из контекстного меню выберите пункты New Entity и далее - Source Code Editor. Из меню File выберите Import, измените текущий каталог на C:\EVO\F1 и дважды щелкните по файлу Cdf.prg для импорта кода метода CreateDataFiles().

В верхней части импортированного кода вы увидите блок текста, который должен выглядеть примерно так:

TEXTBLOCK CDF.PRG

Import DATE: 01/01/99 12:12:12 PM
Эти строки могут быть удалены, чтобы оставить только импортированный код. Сохраните код и давайте посмотрим, что же мы импортировали.

Если взять в качестве примера первую конструкцию IF…ENDIF, то обратите внимание на то, что первые три строки должны читаться как одна. Она разделена на несколько строк лишь для удобочитаемости. Функция DBCreate() используется для создания файла данных. При успешном завершении возвращается TRUE, в противном случае - FALSE. Первым параметром является имя файла, которое может включать и путь к нему (необязательно). Вторым параметром является массив массивов, каждый из подмассивов которого определяет характеристики поля создаваемого файла. В каждом подмассиве содержатся четыре элемента: имя поля - Field Name, которое должно быть не длиннее 8 символов, тип данных - Data Type, размер поля Size, и число десятичных знаков - Decimal Places, которое, очевидно, может относиться только к числовому полю. Некоторые типы данных имеют фиксированный размер полей.
В приведенной ниже таблице дан список различных типов данных с фиксированным размером полей (где это применимо):

Константа
Тип 

Размер 
Десятичные







знаки

‘C’ 

Символьное 


0

‘N’ 

Числовое

‘D’ 

Дата 

8 

0

‘L’ 

Логическое 
1 

0

‘M’ 

Мемо
 
10 

0

‘O’ 

Объект 
10 

0
В функции DBCreate() имеются и другие параметры, полный список которых может быть найден в документации, но в данный момент нам это не нужно. Если функция возвращает FALSE, выводится соответствующее окно с именем файла.

Другие файлы DBF создаются аналогично, поэтому мы перейдем к следующему  разделу, в котором создаются индексные файлы.

Если файл Drivers.dbf создан, он открывается функцией DBUseArea(), а параметр TRUE указывает функции, что файл должен быть открыт в новой рабочей области. При удачном открытии файла вызывается функция DBCreateIndex(), которой передаются два параметра: первый - имя создаваемого индексного файла, второй - ключевое выражение. При первом вызове создается файл с именем DrvName, проиндексированный по полю DriName. При ошибке создания файла выводится сообщение об этом, в противном случае мы переходим к созданию индекса DrivNo по полю KeyNo. И вновь, при ошибке выводится соответствующее сообщение, в противном случае мы готовы к вводу данных.

Первое обращение к DBCreateIndex() создает и открывает индексный файл, однако следующее обращение закрывает все открытые индексные файлы перед началом своей работы. Это означает, что мы должны вызвать функцию DBSetIndex() для повторного открытия первого индексного файла.

Следующая строка показывает использование функции DBAppDelim() для работы с файлом с разделителями (Delimited) Drivers.txt, который использует запятые в качестве разделителя полей. Файл Drivers.txt находится в каталоге C:\EVO\F1 и может быть открыт в любом текстовом редакторе для ознакомления с его содержанием и структурой. Далее идут аналогичные блоки кода, создающие другие индексные файлы, необходимые для работы приложения. В заключение, перед возвратом, осуществляется вызов функции DBCloseAll(), которая закрывает все ранее открытые файлы.

Сохраните код, закройте редактор, скомпилируйте и запустите приложение.

На этот раз запуск приложения займет несколько больше времени, но все равно на экране появится то же окно. Закройте приложение и загляните в каталог C:\EVO\F1. Там вы увидите все созданные файлы. Теперь, после их создания, мы можем продолжить создание окон.

Давайте начнем с создания требуемых серверов данных. Создайте новый модуль с именем "S:Drivers" и создайте сервер, используя файлы Drivers.dbf с индексным файлом DrivNo.ntx и DrivName.ntx, при этом используйте все поля и удалите пути из всех имен файлов. Сохраните сервер и закройте редактор.

Создайте новый модуль с именем "S:Circuit" и создайте сервер, используя файлы Circuit.dbf с индексным файлом CircNo.ntx и CircName.ntx. Вновь используйте все поля и удалите пути из всех имен файлов. Сохраните сервер и закройте редактор.

Создайте новый модуль с именем "S:Teams" и создайте сервер, используя файлы Teams.dbf с индексным файлом TeamNo.ntx и TeamName.ntx. Как и ранее, используйте все поля и удалите пути из всех имен файлов. Сохраните сервер и закройте редактор.

И, наконец, создайте новый модуль с именем "S:Races" и создайте сервер, используя файлы Races.dbf с индексным файлом RaceDate.ntx и RacName.ntx. Вновь используйте все поля и удалите пути из всех имен файлов. Сохраните сервер и закройте редактор.

Теперь у нас есть четыре сервера, скомпилируйте приложение для того, чтобы убедиться в отсутствии ошибок.

Для корректного отображения данных в дальнейшем необходимо обеспечить, чтобы индексом по умолчанию для сервера Drivers был индекс DrivNo, для сервера Circuits - индекс CircNo и TeamNo для сервера Teams. Хотя мы можем установить это все и в редакторе, наилучшим способом, гарантирующим нужный порядок подключения индексов, является создание метода PostInit() для каждого из серверов с жестким подключением индексов.

Для этого откройте в редакторе сервер Drivers, в окне его свойств выберите PostInit и укажите в нем Defined. Сохраните сделанные изменения и закройте редактор. Теперь в правой панели репозитория вы можете увидеть примитив с именем PostInit(). Дважды щелкните по нему мышью для вызова редактора кода и измените код следующим образом:

METHOD PostInit() CLASS Drivers

// Put your PostInit additions here

SELF:SetOrder("DrivNo") 

// Новая строка
RETURN NIL
Создайте такие же методы для серверов Circuits и Teams с вызовами SetOrder() для сервера Circuit -

SELF:SetOrder("CircNo"),

а для сервера Teams -

SELF:SetOrder("TeamNo").

Элемент управления TreeView
Следующее окно, с которым мы будем иметь дело, связано с окном с кнопками в первой панели. Оно будет располагаться в третьей панели, непосредственно под первой. Создайте новый модуль с именем "W:TreeWin", откройте оконный редактор, выберите тип окна DIALOGWINDOW, назовите его TreeWin и нажмите кнопку ОК. Нам нужен элемент TreeView, соответственно, выберите его в палитре элементов управления, поместите его на форму окна и назовите "F1Tree". Перейдите на вкладку General и установите свойство Owner Alignment в значение OA_FULL_SIZE. Это обеспечит установку размера элемента TreeView соответственно размеру окна. Щелкните по окну, чтобы передать ему фокус.

Перейдите на вкладку DlgWindow в свойствах окна. В его свойстве Type укажите TabPage, а Modeless установите в Yes. Перейдите к вкладке Tabs и отключите заголовок окна, установив свойство Caption Bar в No Caption Bar. В завершение установите размеры окна 155х240, а размеры элемента TreeView такими, чтобы он покрывал всю площадь окна.

Сохраните код, закройте редактор, скомпилируйте и запустите приложение.

Последней частью этого раздела будет интеграция окна в приложение. Для этого откройте метод Start() в редакторе кода и добавьте следующую строку после существующих объявлений локальных переменных.

LOCAL oTreeWin AS TreeWin

Кроме того, добавьте следующие строки:

oShowWin:Show( )

oSplit:SetPaneClient(oShowWin, 1)

oTreeWin := TreeWin{oSplit} 

// Новая строка
oTreeWin:Size := oSplit:GetPaneSize(3) 
// Новая строка
oTreeWin:Show( ) 


// Новая строка

oSplit:SetPaneClient(oTreeWin, 3) 
// Новая строка
SELF:Exec( )

RETURN NIL
Сохраните код и сравните этот блок с кодом в предыдущем блоке.

Мы вновь создаем окно, однако в данном случае мы устанавливаем размер окна в зависимость от размера панели 3, вместо того, чтобы устанавливать размер панели в соответствии с размером окна. Затем окно появляется на экране и помещается в третью панель нашего разделенного окна.

Сохраните код, закройте редактор, скомпилируйте и запустите приложение.

В данный момент ничего особенного в окне приложения не видно, поскольку элемент TreeView занимает всю третью панель и имеет такой же белый фон, как и остальная часть SplitWindow. Закройте приложение и давайте добавим немного данных в новую панель.

Сейчас мы должны разобраться с отношениями между данными и панелями. Окно ShowWin, созданное нами ранее, должно обеспечивать различные варианты просмотра различных файлов, однако само по себе оно не отображает никакие данные, следовательно, оно и не должно владеть никаким из серверов. Только что созданное окно TreeWin должно отображать данные, однако оно должно разделять эти данные с двумя другими панелями, которые еще предстоит создать. Общим элементом для всех упомянутых окон является его владелец. Это приводит нас к выводу о том, что окно SplitWindow должно быть владельцем всех серверов.

Вместо того, чтобы переделывать класс SplitWindow, мы попробуем унаследовать класс от него и расширить возможности вновь создаваемого класса. Создайте новый модуль с именем "C:F1SplitWindow", откройте редактор кода и импортируйте файл F1Split.prg в этот модуль. Снова вверху окажется блок текста, который можно удалить, после чего сохраните код и посмотрите на то, что мы импортировали.

Сначала следует объявление класса, демонстрирующее механизм наследования, а также четыре переменные для каждого из серверов. Далее следует метод инициализации Init(), который вызывает метод SUPER:Init(), а затем открывает все четыре сервера. За этим следует вызов метода Close(), который закрывает упомянуты серверы. Кроме того, имеются четыре метода доступа по чтению - ACCESS, которые позволяют использовать серверы из любого другого окна, используя запись вида:

SELF:Owner:<Access Name>

Сохраните код, закройте редактор, перейдите в модуль C:Start и откройте метод Start().

В нем есть два места, где упоминается SplitWindow, в обоих случаях его следует заменить на F1SplitWindow, теперь так будет называться наше новое окно. Сохраните код, закройте редактор, скомпилируйте и запустите приложение.

И вновь ничего не видно, однако теперь мы знаем, что все работает как надо. Закройте приложение и давайте выведем некоторые данные на экран.

Мы определили кнопку Drivers как кнопку по умолчанию в окне ShowWin, соответственно, мы должны создать элемент Drivers TreeView для окна TreeWin по умолчанию. Немного подумав, мы придем к выводу, что это не единственный случай, когда потребуется отображение пилотов (Drivers) в TreeView. Всякий раз, когда пользователь выберет кнопку Drivers, нам вновь потребуется показать этот список. Сие означает, что эта возможность должна быть определена в методе, который может быть вызван в любое нужное время. Перейдите в модуль W:TreeWin, выберите пункт Source Code Editor из меню New Entity и введите следующий код:

METHOD DriversFill( ) CLASS TreeWin

LOCAL oItem AS TreeViewItem
SELF:oDCF1Tree:DeleteAll( )

SELF:Owner:DriversSvr:SetOrder('DriName')

SELF:Owner:DriversSvr:GoTop()

oItem := TreeViewItem{ }

DO WHILE ! SELF:Owner:DriversSvr:Eof

oItem:NameSym := String2Symbol('ROOT ' + SELF:Owner:DriversSvr:KeyNo)

oItem:TextValue := SELF:Owner:DriversSvr:DriName

oItem:Value := SELF:Owner:DriversSvr:KeyNo

SELF:oDCF1Tree:AddItem(#ROOT, oItem)

SELF:Owner:DriversSvr:Skip( )

ENDDO

SELF:Owner:DriversSvr:SetOrder('DrivNo')

RETURN NIL
Сохраните код и давайте посмотрим, что мы сделали. После объявления переменной, которая будет использована для создания элементов TreeViewItem, мы вызываем метод DeleteAll() класса TreeView, поскольку в будущем этот элемент никогда не будет пустым. Далее мы устанавливаем индекс DriName и переходим к первой логической записи. Перед входом в цикл DO WHILE мы создает пустой элемент TreeViewItem. Цикл будет выполняться до достижения конца файла. 

Внутри цикла мы присваиваем значения следующим элементам:

( <NameSym> устанавливается в значение ROOT плюс <KeyNo> из сервера;

( <TextValue> устанавливается в значение <DriName> из сервера;

( <Value> устанавливается в значение <KeyNo>, также из сервера.

После этого элемент Item (ветвь дерева) добавляется к элементу TreeView со специальным  параметром #ROOT. Это значение является специальным потому, что указывает на то, что это корневой, а не вложенный элемент некоторого уже существующего корневого элемента. В конце цикла используется метод Skip() и мы либо начинаем повторное прохождение цикла, либо завершаем работу, в зависимости от текущего положения в сервере.

Закройте редактор, откройте элемент TreeView в оконном редакторе, щелкните по свойству PostInit, чтобы открыть редактор кода, и измените существующий код следующим образом:

METHOD PostInit(oParent,uExtra) CLASS TreeWin

//Put your PostInit additions here

SELF:DriversFill( )

RETURN NIL
Очевидно, это вызов только что созданного нами метода, соответственно, при открытии окна элемент TreeView заполнен данными и они соответствуют выбранной кнопке. Сохраните код, закройте редактор, скомпилируйте и запустите приложение.

Наконец, у нас появились некоторые данные на экране (рис. 21.6), при этом кажется, что панели 1 и 3 соединены. Выбор любого из элементов переместит фокус, как и ожидалось, к этому элементу, однако это только один из вариантов представления данных, а они должны изменяться в зависимости от выбранной в данный момент кнопки. Для этого нам нужна два других способа заполнения окна данными. Закройте приложение и откройте метод DriversFill() в редакторе кода.

Управление отображением элемента TreeView
Скопируйте весь код в буфер обмена, закройте редактор, откройте его вновь из контекстного меню, вставьте код из буфера и измените первую строку на 

METHOD CircuitFill() CLASS TreeWin

Сохраните, на всякий случай, новый код. Теперь измените код так, как показано ниже (необходимые изменения подчеркнуты):

METHOD CircuitFill( ) CLASS TreeWin

LOCAL oItem AS TreeViewItem
SELF:oDCF1Tree:DeleteAll()

SELF:Owner:CircuitSvr:SetOrder('CircName')

SELF:Owner:CircuitSvr:GoTop()

oItem := TreeViewItem{ }
DO WHILE ! SELF:Owner:CircuitSvr:Eof

oItem:NameSym := String2Symbol('ROOT ' + SELF:Owner:CircuitSvr:KeyNo)

oItem:TextValue := SELF:Owner:CircuitSvr:CirName
oItem:Value := SELF:Owner:CircuitSvr:KeyNo

SELF:oDCF1Tree:AddItem(#ROOT, oItem)

SELF:Owner:CircuitSvr:Skip()

ENDDO

SELF:Owner:CircuitSvr:SetOrder('CircNo')

RETURN NIL
Сохраните код, закройте редактор, откройте метод PostInit() и замените вызов DriversFill() на CircuitFill() для того, чтобы можно было опробовать метод. Сохраните код, закройте редактор, скомпилируйте и запустите приложение. На этот раз мы должны увидеть в панели TreeView все наименования трасс. Закройте приложения и давайте добавим последний метод.

Откройте в редакторе метод DriversFill(). Скопируйте весь код в буфер обмена, закройте редактор, откройте его вновь из контекстного меню, вставьте код из буфера и измените его следующим образом (необходимые изменения подчеркнуты):

METHOD TeamsFill( ) CLASS TreeWin

LOCAL oItem AS TreeViewItem
SELF:oDCF1Tree:DeleteAll()

SELF:Owner:TeamsSvr:SetOrder('TeamName')

SELF:Owner:TeamsSvr:GoTop()

oItem := TreeViewItem{ }
DO WHILE ! SELF:Owner:TeamsSvr:Eof

oItem:NameSym := String2Symbol('ROOT ' + SELF:Owner:TeamsSvr:KeyNo)

oItem:TextValue := SELF:Owner:TeamsSvr:TeaName
oItem:Value := SELF:Owner:TeamsSvr:KeyNo

SELF:oDCF1Tree:AddItem(#ROOT, oItem)

SELF:Owner:TeamsSvr:Skip()

ENDDO
SELF:Owner:TeamsSvr:SetOrder('TeamNo')

RETURN NIL
Сохраните код, закройте редактор, откройте метод PostInit() в редакторе и замените вызов CircuitFill() на TeamsFill() для того, чтобы можно было опробовать метод. Сохраните код, закройте редактор, скомпилируйте и запустите приложение. На этот раз мы должны увидеть в панели TreeView все наименования команд гонщиков. Закройте приложение и давайте добавим в него еще некоторые возможности.

Первое, что нам нужно изменить, это метод PostInit(), где вместо SELF:TeamsFill() нужно снова вставить SELF:DriversFill() для того, чтобы данные соответствовали кнопке, выбранной по умолчанию. Далее мы должны найти способ передать сообщение о щелчке мыши из окна ShowWin в первой панели окну TreeWin в третьей панели. Для этого перейдите в модуль W:ShowWin и откройте окно ShowWin в оконном редакторе. Перейдите на вкладку Control Events свойств окна, щелкните по свойству ButtonClick и введите следующий код:

METHOD ButtonClick(oControlEvent) CLASS ShowWin

LOCAL oControl AS Control
oControl := IIf(oControlEvent == NULL_OBJECT, NULL_OBJECT, oControlEvent:Control)

SUPER:ButtonClick(oControlEvent)

//Put your changes here

SELF:Owner:ButtonHandler(oControl:NameSym)

RETURN NIL
В новой строке осуществляется вызов метода ButtonHandler() в окне владельца и в качестве параметра ему передается символьное имя <NameSym> элемента управления, который вызывает этот метод. Сохраните код, закройте редактор и перейдите в модуль C:F1SplitWindow, откройте редактор кода и введите следующее:

METHOD ButtonHandler( symButton ) CLASS F1SplitWindow
DO CASE

CASE symButton = #rbDrivers

SELF:GetPaneClient(3):DriversFill()

CASE symButton = #rbTeams

SELF:GetPaneClient(3):TeamsFill()

CASE symButton = #rbCircuits

SELF:GetPaneClient(3):CircuitFill()

ENDCASE

RETURN NIL
Сохраните код и посмотрите, как это работает. При вызове этого метода из уже рассмотренного метода ButtonClick(), переданное символьное имя обрабатывается в цикле DO CASE…ENDCASE. В зависимости от результатов обработки выбирается соответствующий метод представления данных для объекта, возвращаемого методом GetPaneSize(), т.е. в данном случае - для окна TreeWin.

Сохраните код, закройте редактор, скомпилируйте и запустите приложение. Теперь при выборе любой из кнопок должно изменяться представление данных в окне TreeWin. По окончании тестирования закройте приложение и можно перейти к следующей части работы. 

Вкладки

Следующим шагом будет размещение окна TreeWin, находящегося сейчас в третьей панели, на вкладке в той же панели с тем, чтобы потом добавить вторую вкладку, использующую то же самое "жизненное пространство".

Для этого нужно создать новое окно. Создайте новый модуль, назовите его W:TabWin, откройте оконный редактор, выберите тип окна DIALOGWINDOW и назовите окно TabWin. Перейдите на вкладку DlgWindow в свойствах окна, установите его тип - Type в TabPage, свойству Modeless присвойте значение Yes и отключите заголовок окна, установив его свойство Caption Bar в значение No Caption Bar.

Из палитры инструментов выберите элемент управления Tab Control и расположите его в верхнем левом углу окна. Свойству Name на вкладке свойств присвойте значение "tcView", свойству CurrentTab:Caption - значение "View Detail", а свойству CurrentTab:Name - "TreeWin". Установите ширину 155, высоту - 230, левый отступ - 3, верхний отступ - 3. Выравнивание в окне-владельце (Owner Alignment) установите в значение OA_FULL_SIZE.

Важно отметить, что все свойства относятся к элементу управления Tab, за исключением двух, с которых начинается Current Tab: - они относятся к той вкладке, которая в данный момент имеет фокус. Теперь щелкните по вкладке правой кнопкой мыши и выберите Add Page (Добавить вкладку) из контекстного меню. Теперь мы увидим наверху вторую вкладку, однако она пока не имеет фокуса, соответственно, щелкните по ней левой кнопкой мыши и вкладка переместится на передний план. В окне ее свойств перейдите на вкладку General, измените свойство Current Tab:Caption на "Race Season" (Год гонки) , а Current Tab:Name - на "SeasonWin".

Последнее, что нам осталось сделать, это щелкнуть по окну, чтобы передать ему фокус и на вкладке свойств окна DlgWindow установить требуемые размеры: ширину - 155 и высоту - 235.

 Сохраните код, закройте оконный редактор и откройте метод Start(). В начале метода замените строку

LOCAL oTreeWin AS TreeWin

на строку

LOCAL oTabWin AS TabWin
Кроме того, замените строки

oTreeWin:= TreeWin{oSplit}

oTreeWin:Size := oSplit:GetPaneSize(3)

oTreeWin:Show( )

oSplit:SetPaneClient(oTreeWin, 3)
строками

oTabWin:= TabWin{oSplit}

oTabWin:Size := oSplit:GetPaneSize(3)

oTabWin:Show( )

oSplit:SetPaneClient(oTabWin, 3)
после чего сохраните код и посмотрите, что из этого получится при запуске приложения. Вместо создания окна TreeWin в третьей панели будет создано окно TabWin. Оно владеет элементом управления Tab, для которого окно TreeWin является одной из его страниц. Все это означает, что в созданной нами цепочке появилось дополнительное окно, но какое практическое значение это имеет для нас?

Закройте редактор, перейдите в модуль W:TreeWin и откройте метод DriversFill(). Первым  местом, в котором скажутся сделанные изменения, является строка

SELF:Owner:DriversSvr:SetOrder('DrivName')
потому что теперь появилась ошибка в определении окна-владельца. Для того, чтобы эта строка заработала, ее нужно изменить следующим образом:

SELF:Owner:Owner:DriversSvr:SetOrder('DrivName')
В доказательство этого мы изменим метод (необходимые изменения подчеркнуты)

METHOD DriversFill( ) CLASS TreeWin

LOCAL oItem AS TreeViewItem
SELF:oDCF1Tree:DeleteAll()

SELF:Owner:Owner:DriversSvr:SetOrder('DrivName')

SELF:Owner:Owner:DriversSvr:GoTop()

oItem := TreeViewItem{ }

DO WHILE ! SELF:Owner:Owner:DriversSvr:Eof

oItem:NameSym := String2Symbol( 'ROOT ' + SELF:Owner:Owner:DriversSvr:KeyNo)

oItem:TextValue := SELF:Owner:Owner:DriversSvr:DriName

oItem:Value := SELF:Owner:Owner:DriversSvr:KeyNo

SELF:oDCF1Tree:AddItem(#ROOT, oItem)

SELF:Owner:Owner:DriversSvr:Skip()

ENDDO

SELF:Owner:Owner:DriverSvr:SetOrder('DrivNo')

RETURN NIL
Сохраните код, закройте редактор, скомпилируйте и запустите приложение.

Прежде, чем двигаться дальше, обратите внимание на количество вкладок в третьей панели (Рис. 21.7). Там только одна, но мы создавали две -где же вторая вкладка? Ответ заключается в том, что нами не был создан объект (предпочтительнее окно, иначе возникнет ошибка) с тем же именем SeasonWin. Поскольку такого объекта не существует, вкладка не создается.

Последнее, что нам осталось опробовать здесь, это работа кнопок. Это несомненно при ведет к аварийному завершению работы приложения, поскольку мы переработали только один метод - DriversFill() с учетом существования дополнительного окна. Нажмите кнопку Abort в окне с сообщением об ошибке, это вызовет сообщение о завершении работы приложения -Execution Finished и теперь мы можем закрыть приложение.

Первое, что нужно сделать - исключить крах приложения при нажатии на одну из кнопок в окне ShowWin. Как мы уже видели, методы Fill…() требуют дополнительного указания владельца, однако это не вся проблема. Наш способ обработки нажатий на кнопки заключается в посылке сообщения вверх по цепочке -

SELF:Owner:ButtonHandler(oControl:NameSym)
Это сработает, поскольку владелец окна ShowWin не изменился и метод ButtonHandler() в окне F1SplitWindow запустится, однако именно здесь и возникает проблема. Предположим, нажата кнопка Drivers. Тогда будет вызвана следующая строка -

SELF:GetPaneClient(3):DriversFill()
Между тем, метод DriversFill() более не является методом клиента окна в третьей панели. Теперь это метод клиента, который, в свою очередь, сам является клиентом в третьей панели. Сейчас нам нужен способ передачи сообщения дальше вверх по цепочке. Самым простым способом будет перейти в модуль W:TabWin, создайте новый примитив и введите следующий код:

METHOD CircuitFill( ) CLASS TabWin

SELF:oTPTreeWin:CircuitFill()

RETURN NIL
METHOD DriversFill( ) CLASS TabWin

SELF:oTPTreeWin:DriversFill()

RETURN NIL
METHOD TeamsFill( ) CLASS TabWin

SELF:oTPTreeWin:TeamsFill()

RETURN NIL
Сохраните код и давайте разберем, что мы сделали. Метод, называемый <…>Fill() в окне TabWin вызывает тот же метод для объекта, содержащегося в <oTPTreeWin>. Это защищенная переменная, в которой содержится объект TreeWin, завершающий цепочку и с ее помощью сообщение передается надлежащему окну. Закройте редактор и давайте решим, что делать дальше.

Теперь у нас есть правильный метод, вызываемый при нажатии на кнопку, однако, как мы видели в методе DriversFill(), сервер, который мы пытаемся использовать, находится на один шаг дальше, чем нужно. Мы можем доработать оба оставшихся метода аналогичным способом. Альтернативным решением будет создание некоей "прокладки", которая будет направлять сообщения нужному окну.

Мы уже разрешили эту проблему одним способом, давайте попробуем решить ее иначе. Перейдите в модуль W:TabWin, создайте новый примитив и введите следующий код:

ACCESS TeamsSvr CLASS TabWin

RETURN SELF:Owner:TeamsSvr
ACCESS CircuitSvr CLASS TabWin

RETURN SELF:Owner:CircuitSvr
Это работает как описано ранее, поэтому не требует дополнительных объяснений. Сохраните код, закройте редактор, скомпилируйте и запустите приложение.

Теперь блок кнопок работает, как и ожидалось. По окончании тестирования закройте приложение и мы двинемся дальше.

Теперь перейдем ко второй вкладке, которая до сих пор не видна. Для работы этой вкладки нам понадобятся два новых сервера:

· создайте новый модуль "S:Engine", откройте редактор серверов и создайте сервер из файлов Engine.dbf и Engine.ntx. Сохраните сервер и закройте редактор.

· создайте новый модуль "S:RaceYr", откройте редактор серверов и создайте сервер из файлов RaceYr.dbf и RaceYr.ntx. Сохраните сервер и закройте редактор.

Скомпилируйте приложение, чтобы обеспечить доступ к серверам.

Создание следующего окна мы начнем с создания нового модуля "W:RaceSeason". Откройте редактор кода и вставьте следующее:

CLASS RaceSeason INHERIT TreeWin
METHOD PostInit(oParent,uExtra) CLASS RaceSeason

SELF:RaceYrFill( )
RETURN NIL
METHOD RaceYrFill( ) CLASS RaceSeason

RETURN NIL
Сохраните код и посмотрите, что мы сделали. Мы создали новый класс RaceSeason, который обладает всеми характеристиками предка - класса TreeWin. Это означает, что мы можем использовать методы CircuitFill() или DriversFill(). В настоящее время этот класс является точно таким же, как и его предок, пока это бессмысленно, однако, мы получаем основу для работы. При запуске метода Init() в действительности будет запущен аналогичный метод класса TreeWin, поскольку в настоящий момент метод Init() класса Race Season еще не написан. Однако, когда метод Init() вызовет метод PostInit(), будет вызван именно тот метод, который только что написан нами. При этом будет вызван метод RaceYrFill(), который в настоящий момент не делает ничего, однако позднее мы вернемся к нему. Сейчас закройте редактор, скомпилируйте приложение с тем, чтобы можно было встроить в него наше новое окно.

Перейдите в модуль W:TabWin и откройте окно TabWin. Щелкните по вкладке RaceSeason в элементе управления Tab, откройте вкладку General окна свойств и перейдите к свойству Current Tab:Name. В данный момент имя указано как SeasonWin. Измените его на "RaceSeason", сохраните окно и закройте редактор.

Откройте метод Init() в редакторе кода. Посмотрите на строку 13, где создается первая вкладка, и строку 14, где она добавляется к элементу управления TabControl. Нам не хватает создания второй страницы, хотя имена уже и соответствуют объекту. Причиной этого является то, что при генерации кода оконный редактор ищет соответствующее окно. Если бы он искал класс, то мог бы найти сервер или меню, что повлекло бы за собой крах приложения во время выполнения.

Закройте редактор и вновь откройте окно, чтобы исправить существующее положение. Перейдите на вкладку DlgWindow в окне свойств, щелкните по свойству Class Declaration и измените в редакторе код следующим образом:

CLASS TabWin INHERIT DIALOGWINDOW

PROTECT oDCtcView AS TABCONTROL

PROTECT oTPTREEWIN AS TREEWIN

//{{%UC%}} USER CODE STARTS HERE (do NOT remove this line)
PROTECT oTPRaceSeason AS RaceSeason
В этой переменной содержится наша новая страница вкладки. Сохраните код, закройте редактор, перейдите в конец страницы свойств окна DlgWindow, откройте свойство PostInit и измените код следующим образом:

METHOD PostInit(oParent,uExtra) CLASS TabWin

//Put your PostInit additions here

SELF:oTPRaceSeason := RaceSeason{ SELF }

SELF:oDCtcView:AppendTab(#RaceSeason, 'Race Season', SELF:oTPRaceSeason, 0)

RETURN NIL
Это весьма близко к тому, что мы видели ранее в строках 13 и 14 метода Init(). Сделанные нами изменения обеспечивают использование окна RaceSeason. Сохраните код, закройте редактор, закройте оконный редактор, скомпилируйте и запустите приложение.

Теперь отображаются обе вкладки, щелкните по вкладке RaceSeason. Поскольку нами не написан метод обработки данных для нее, неудивительно, что окно во вкладке пусто (рис. 21.8), однако, как уже упоминалось, мы к этому еще вернемся. Находясь здесь, щелкните по одной из кнопок, а затем перейдите на вкладку View Details. Вы увидите, что содержание окна на вкладке изменилось, независимо от того, что фокус имела другая вкладка, что подтверждает работу системы нотификации. По окончании тестирования закройте приложение и давайте попробуем ввести некоторые данные в наше новое окно.

И вновь к TreeViews

Первое, что нужно принять во внимание, это то, что нам требуются методы доступа к ранее созданным серверам. Перейдите в модуль C:F1SplitWindow и откройте объявление класса в редакторе кода. Дополните объявление класса следующими строками:

PROTECT oEngine AS Engine

PROTECT oRaceYr AS RaceYr
Сохраните код, закройте редактор, откройте метод Init() и дополните его следующими строками непосредственно перед возвратом из метода:

SELF:oEngine := Engine{ }

SELF:oRaceYr := RaceYr{ }
Сохраните код, закройте редактор, откройте метод Close() и дополните его следующими строками непосредственно перед возвратом из метода:

SELF:oEngine:Close()

SELF:oRaceYr:Close()
Сохраните код, закройте редактор, создайте новый примитив и введите следующий код:

ACCESS RaceYrSvr CLASS F1SplitWindow

RETURN SELF:oRaceYr
ACCESS EngineSvr CLASS F1SplitWindow

RETURN SELF:oEngine
Сохраните код, закройте редактор, скомпилируйте и запустите приложение.

Теперь, когда у нас есть способ получения данных, давайте "напишем" метод заполнения окна. Перейдите в модуль W:RaceSeason, откройте метод RaceYrFill(), удалите весь имеющийся код и импортируйте файл Ryf.prg, это сохранит нам немало времени! Сохраните код и посмотрите, что делает этот код.

Внешний цикл DO WHILE…ENDDO создает корневые элементы, как и ранее, с той лишь разницей, что корневыми элементами являются названия сезонов (в данном контексте - годов - прим. пер.) и каждый из них упоминается лишь единожды в году.

Внутри каждого из годов мы проходим по записям, добавляя команды, однако в сервере мы имеем лишь <TeamRef>, соответственно, нам необходимо заглянуть в сервер Teams и получить нужный текст оттуда. Единственное, что нам нужно принять во внимание, так это то, что наименование команды может повторяться несколько раз, для этого еще один внутренний цикл проходит по записям файла при неизменном значении <TeamRef>.

Еще одним важным моментом является присвоение значения символьного имени <oItem:NameSym>. Это может быть любое уникальное значение и в данном случае мы создаем ключ, который далее будет иметь несколько применений. Первая часть ключа включает предыдущий элемент, а вторая часть включает в себя собственный <KeyNo>. Такая комбинация делает этот ключ уникальным. Сохраните код, закройте редактор, скомпилируйте и запустите приложение.

Теперь при выборе вкладки Race Season мы получим год в свернутом виде с символом развертывания слева от него. Щелкните по значку плюса, свернутый до этого пункт раскроется и мы увидим список участвовавших команд, каждая из которых упоминается только один раз (рис. 21.9). В нашем представлении имеются только два уровня, однако число их может быть легко увеличено.

Еще нам нужно помнить, что все время, пока элемент TreeView заполняется, окна нет на экране. Мы можем исправить это положение с помощью подручных средств, реагируя соответствующим образом на щелчок мыши в методе TreeViewSelectionChanged() для того, чтобы разворачивать имеющиеся ветви дерева и добавлять новые. По окончании тестирования закройте приложение и мы добавим недостающую часть.

В модуле W:RaceSeason создайте новый примитив и, используя опцию импортирования, импортируйте файл Tvsc_rs.prg из каталога C:\EVO\F1. Удалите текстовый блок так, чтобы первой строкой метода стала строка:

METHOD TreeViewSelectionChanged(oTreeViewSelectionEvent) CLASS RaceSeason
В первых нескольких строках объявляются переменные, две из которых - элементы TreeViewItems и три - строки. Затем следует вызов метода SUPER:TreeViewSelectionChanged(), что гарантирует работу метода по умолчанию. Объект события, передаваемый в вызов, имеет указатели как на предыдущий выделенный элемент, так и на элемент, на который осуществляется переход. На следующей строке, первой строке реального кода, используется указатель на новый выделенный элемент.

В следующей строку выполняется проверка, не является ли выделенный элемент корневым. Проверка нужна потому, что если элемент является корневым, то все вложенные элементы и так уже выведены в окне и ничего дополнительно делать не нужно. Предполагая, что это не корневой элемент, мы пробуем получить первый дочерний элемент вновь избранного элемента. Если возвращается ненулевой объект, это значит, что элемент выделялся ранее и дочерние элементы уже имеются в окне. Если же возвращен NULL_OBJECT, то элемент выбран впервые и необходимо создавать дочерние элементы.

Сначала мы получаем значения Season и Team из выделенного элемента и осуществляем их поиск с использованием метода Seek() в соответствующих серверах. Далее необходимо убедиться, что ведущим индексом для сервера Drivers является индекс DrivNo, чтобы можно было использовать Seek для поиска и в этом сервере. Затем можно начинать выполнение цикла DO WHILE … ENDDO, который непосредственно добавляет дочерние элементы в дерево. Это похоже на метод, использованный ранее, однако в данном случае используется сервер Drivers. В завершение мы вызываем метод EnsureVisible(), передавая ему добавляемый элемент. Это обеспечит развертывание ветви дерева и гарантирует то, что данный элемент будет виден в окне. Сохраните код, закройте редактор, скомпилируйте и запустите приложение для тестирования.

При появлении окна на экране перейдите на вкладку Race Season и разверните элемент "1998 год".

Теперь, если вы выберите любую команду, вы получите список пилотов в виде дочерних элементов наименования команды (рис. 21.10). для проверки попробуйте перейти к другой команде, а затем возвратиться к первой, чтобы убедиться, что количество пилотов не изменилось. На этом завершим раздел и закроем приложение, чтобы двинуться далее.

Элемент ListView

Однако, довольно говорить о TreeView, давайте перейдем к списочному элементу ListView. Создайте новый модуль, назовите его "W:AllRaces", откройте оконный редактор, выберите тип окна DATAWINDOW и назовите его "AllRaces". Используйте возможность автораскладки и сервер Races, используя все имеющиеся поля и измените форму представления на табличную (Browse View). Перейдите на вкладку DataWindow окна свойств щелкните по свойству Browser Inherits From. При этом развернется комбинированный список, в котором нам предоставлены три возможности. Выберите DataListView из имеющего списка и перейдите к свойству Columns Inherit From. В этом списке необходимо выбрать ListViewColumn. На данном этапе это все, что нам нужно, сохраните окно и закройте редактор.

Далее необходимо встроить это окно в наше приложение. Перейдите в модуль C:Start, откройте метод Start() и добавьте еще одну строку в объявление переменных -

LOCAL oAllRaces AS AllRaces

Теперь перейдите в самый конец метода и внесите следующие изменения:

oSplit:SetPaneClient(oTabWin, 3)

oAllRaces := AllRaces{oSplit} 

// Новая строка
oAllRaces:Size := oSplit:GetPaneSize(2) 
// Новая строка
oSplit:SetPaneClient(oAllRaces, 2) 
// Новая строка
oAllRaces:Show( ) 


// Новая строка

SELF:Exec( )
Здесь нет ничего нового, соответственно, сохраните код, закройте редактор, скомпилируйте и запустите приложение.

Теперь мы видим списочный элемент ListView во второй панели (рис. 21.11), как и ожидалось, однако вместо пилотов, трасс и команд выведены лишь их ссылочные номера, поэтому такой список не очень полезен. В обычном броузере мы использовали бы свойства Block и BlockOwner для отображения нормальных данных в колонках, в DataListView мы не можем использовать этот способ. Вместо этого мы переопределим метод FieldGet() для класса DataListView. Закройте приложение.

Имена вместо номеров

Перейдите в модуль W:AllRaces, создайте новый примитив и введите код:

CLASS AllRacesListView INHERIT DataListView
Сохраните код, закройте редактор и скомпилируйте приложение. Затем откройте окно AllRaces в оконном редакторе, перейдите на вкладку DataWindow в окне его свойств и выберите свойство Browser Inherits From. В появившемся списке вы теперь увидите дополнительную возможность наследования - от класса AllRacesListView. Выберите эту опцию. Сохраните код, закройте редактор, скомпилируйте и запустите приложение.

Сейчас вы не увидите никаких изменений в поведении приложения. Однако, вскоре все изменится.

Создайте новый примитив и введите следующий код:

METHOD FIELDGET(symValue) CLASS AllRacesListView

LOCAL uRet AS USUAL
uRet := SUPER:FIELDGET(symValue)

DO CASE

CASE symValue = #CIRNAME

IF SELF:Owner:Owner:Owner:CircuitSvr:Seek(uRet)

uRet := SELF:Owner:Owner:Owner:CircuitSvr:CirName

ENDIF

ENDCASE

RETURN uRet
Сохраните код и посмотрите, что он делает. Метод FieldGet() будет вызываться автоматически всякий раз, когда элементу ListView потребуются данные и в метод будет передан параметр - символьное имя колонки, которую требуется заполнить данными.

В первой строке объявляется переменная, в которой будет содержаться возвращаемое значение, однако она объявлена как переменная типа USUAL. Это сделано потому, что значение, возвращаемое SUPER:FieldGet(), будет иметь тот же тип данных, что и соответствующее поле в таблице, следовательно, оно может иметь строковый, числовой, логический, дата, мемо и даже другой объект, поэтому следует использовать тип данных USUAL.

После вызова метода SUPER начинается выполнения условной конструкции DO CASE. В данный момент это не очень нужно, однако впоследствии мы добавим и другие условия CASE. На данном этапе мы сосредоточим усилия на колонке CirName. Если упомянутый метод вызывается для заполнения колонки CirName, <symValue> примет значение "#CIRNAME", а проверка условия CASE возвратит TRUE. В этом случае мы перейдем к выполнению вложенной условной конструкции IF. Здесь выполняется поиск (Seek()) значения, возвращаемого из вызова SUPER в сервере CircuitSvr, котором владеет окно F1SplitWindow. 

Обратите внимание на цепочку сообщений и вы увидите на одно обращение к владельцу Owner больше, чем могли бы ожидать.

SELF:Owner:Owner:Owner:CircuitSvr:Seek(uRet)
Рассматривая эту строку слева направо, можно отметить, что в данном случае SELF представляет собой класс AllRacesListView и, казалось бы, его владельцем должно быть окно данных AllRaces DataWindow, однако это не так. Вместо этого владельцем является класс __FormFrame - компонент DataWindow. Соответственно, в данном случае необходимо обращение вида SELF:Owner:Owner, чтобы получить доступ к окну данных, которым, в свою очередь, владеет окно F1SplitWindow. Именно там мы находим метод доступа по чтению CircuitSvr Access, с помощью которого организуется работа метода Seek(). Все сказанное действительно только для элемента DataListView. "Нормальный" элемент ListView, помещенный в окно данных, будет иметь ожидавшуюся цепь владения.

Если поиск успешен, мы получаем <CirName> от сервера и возвращаем полученное значение, в противном случае никаких изменений не вносится и возвращаемое значение остается тем же, что получено после обращения к SUPER.

Время теста. Закройте редактор, скомпилируйте приложение. Если все сделано правильно, в колонке Circuit Names должны быть указаны наименования трасс вместо их номеров (рис. 21.12).

По окончании тестирования закройте приложение и добавим код для других колонок, требующих обработки.

Откройте метод AllRacesListView:FieldGet() и добавьте CASE-конструкции для обработки наименований команд Team Names и имен пилотов - Driver Names.

METHOD FIELDGET(symValue) CLASS AllRacesListView

LOCAL uRet AS USUAL
uRet := SUPER:FIELDGET(symValue)

DO CASE

CASE symValue = #CIRNAME

IF SELF:Owner:Owner:Owner:CircuitSvr:Seek(uRet)

uRet := SELF:Owner:Owner:Owner:CircuitSvr:CirName

ENDIF

CASE symValue = #TEANAME 




// Новая конструкция CASE

IF SELF:Owner:Owner:Owner:TeamsSvr:Seek(uRet)

uRet := SELF:Owner:Owner:Owner:TeamsSvr:TeaName

ENDIF

CASE symValue = #DRINAME 




// Новая конструкция CASE

IF SELF:Owner:Owner:Owner:DriversSvr:Seek(uRet)

uRet := SELF:Owner:Owner:Owner:DriversSvr:DriName

ENDIF

ENDCASE
RETURN uRet
Новые конструкции используют тот же подход, что и рассмотренная ранее, поэтому не требуют дополнительных разъяснений. Сохраните код, закройте редактор, скомпилируйте и запустите приложение.

Если все идет по плану, то теперь во всех колонках должны быть выведены соответствующие имена и названия (рис. 21.13). По завершении тестирования закройте приложение и мы может перейти к следующей части.

Вызов ListView из элемента TreeView
В этой части мы перейдем к вводу данных в четвертую панель. Начните с создания нового модуля, назовите его "W:SelectedRaces", откройте оконный редактор, выберите тип окна DATAWINDOW, измените его имя на "SelectedRaces" и нажмите кнопку ОК.

Из палитры инструментов выберите элемент ListView, расположите его в левом верхнем углу окна, установите его размеры  330х140 и присвойте ему имя (Name) - "lvRaces". Вновь выделите окно, установите его размеры равными 335х144 и перейдите к вкладке DataWindow в окне его свойств. Там же установите свойство Data Server в значение Races и сохраните окно. Теперь перейдите к свойству PostInit, откройте его и измените существующий код следующим образом:

METHOD PostInit(oWindow,iCtlID,oServer,uExtra) CLASS SelectedRaces

//Put your PostInit additions here

LOCAL oHyperLabel AS HyperLabel

LOCAL oColumn AS ListViewColumn
oHyperLabel := HyperLabel{#RaceDate, 'Date'}

oColumn := ListViewColumn{11, oHyperLabel}

SELF:oDClvRaces:AddColumn(oColumn)

oHyperLabel := HyperLabel{#DriName, 'Driver'}

oColumn := ListViewColumn{25, oHyperLabel}

SELF:oDClvRaces:AddColumn(oColumn)

RETURN NIL
Сохраните код и посмотрите, как он работает.

Сначала мы объявили две переменные, которые будут содержать создаваемые нами объекты. Первый  из них - объект HyperLabel, содержащий символьное имя и тип данных для колонки. Этот объект передается для создания объекта колонки - ListViewColumn вместе с требуемым размером колонки. В завершение колонка добавляется в объект ListView, расположенный в окне. То же самое делается и для другой колонки. Закройте редакторы.

Следующий шаг состоит в обеспечении способа ввода данных в колонки. Создайте новый примитив, откройте редактор кода и введите следующий код:

METHOD LVCircuitFill( ) CLASS SelectedRaces

? "In Fill Method"

? "------------------"

RETURN NIL
Это просто метод трассировки. Сохраните код, закройте редактор. Позднее мы решим, когда именно этот метод должен быть вызван. Со временем мы заполним ListView именами пилотов, выступавших на выбранных нами трассах и при выборе новой трассы список должен изменяться. Это означает, что мы должны каким-то образом реагировать на изменение выбора трассы в элементе TreeView. Перейдите в модуль W:TreeWin и откройте окно TreeWin в оконном редакторе. Перейдите на вкладку TreeView Events в окне его свойств, откройте свойство TreeViewSelChanged и измените код так:

METHOD TreeViewSelectionChanged(oTreeViewSelectionEvent) CLASS TreeWin

SUPER:TreeViewSelectionChanged(oTreeViewSelectionEvent)

//Put your changes here

? "In Selection Method"

? "-------------------------"

RETURN NIL
Это тоже метод трассировки. Сохраните код, закройте редактор. Сейчас нам нужен способ вывода окна на экран. Как обычно, возвратимся в метод Start(). В конец объявления локальных переменных добавьте строку:

LOCAL oSelectedRaces AS SelectedRaces
А теперь в конце метода добавьте следующие строки:

oAllRaces:Show()

oSelectedRaces := SelectedRaces{oSplit} 

// Новая строка
oSelectedRaces:Size := oSplit:GetPaneSize(4) 
// Новая строка
oSplit:SetPaneClient(oSelectedRaces, 4) 

// Новая строка
oSelectedRaces:Show() 



// Новая строка
SELF:Exec()
Снова ничего нового, посему сохраните код, закройте редактор, скомпилируйте и запустите приложение.

После запуска приложения вы увидите две колонки, добавленные нами к новому окну, но у нас до сих пор нет окна трассировки. Мы хотим получить ответную реакцию на щелчок мыши, соответственно, нажмите кнопку Circuits. И вновь никаких следов окна трассировки. Щелкните по наименованию трассы в окне TreeView и появится окно трассировки, из которого видно, что у нас имеется событие выбора (рис. 21.14).

Теперь нажмите любую из оставшихся кнопок, а затем выберите какой-либо элемент в TreeView. И вновь в нас есть событие выбора. для проведения последнего теста перейдите на вкладку Race Seasons, выделите какой-либо элемент и вновь вы увидите то же окно. По окончании тестирования закройте приложение, чтобы мы могли отфильтровать некоторые их упомянутых событий.

Откройте метод TreeViewSelectionChanged() в модуле W:TreeWin и измените его код:

METHOD TreeViewSelectionChanged(oTreeViewSelectionEvent)CLASS TreeWin

SUPER:TreeViewSelectionChanged(oTreeViewSelectionEvent)
IF SELF:Owner:Owner:GetPaneClient(1):AcCircuit:Pressed 
// Новая строка
? "In Selection Method"

? "-------------------------"

ENDIF 









// Новая строка

RETURN NIL
Сохраните код и посмотрите на конструкцию IF … ENDIF. Часть, которая еще не имеет смысла, это AcCircuit, поскольку это должен быть ACCESS кнопки Circuits окна в первой панели. Эта строка кода проверяет свойство кнопки Pushed (Нажата) и возвращает TRUE, если это так. Закройте редактор, перейдите в модуль W:ShowWin, вновь откройте редактор и введите следующее:

ACCESS AcCircuit( ) CLASS ShowWin

RETURN SELF:oCCrbCircuits
Это метод ACCESS, который возвращает TRUE, если кнопка Circuits нажата. Сохраните код, закройте редактор, скомпилируйте и запустите приложение.

После запуска приложения выберите элемент в Drivers TreeView. На этот раз не должно быть сообщения о событии. Это следствие того, что кнопка Circuits не нажата. Нажмите кнопку Teams, выберите элемент в TreeView и вновь мы не должны получить никакого трассировочного сообщения.

Теперь нажмите кнопку Circuits, выберите элемент в TreeView и мы увидим первое трассировочное сообщение. Кажется, что все работает как и должно, однако перейдите на вкладку Race Season и выберите элемент в дереве. Это вновь повлечет за собой появление трассировочного сообщения, следовательно, нам нужно разобраться, почему это происходит и прекратить это безобразие.

По окончании тестирования закройте приложение, перейдите в модуль W:RaceSeason и откройте метод TreeViewSelectionChanged().

Это тот метод, который выполняется, когда вы выбираете элемент в дереве вкладки RaceSeason. Наша проблема порождается вызовом метода SUPER:TreeViewSelectionChanged(). Причина кроется в том, что классом SUPER в данном случае является класс TreeWin, а именно в него мы поместили код для трассировки и позднее заполним элемент ListView. Это означает, что нам нужен другой фильтр для метода трассировки в классе TreeWin, который предотвращал бы запуск метода, если фокус передан во вкладку Race Season.

Закройте редактор, перейдите в модуль W:TreeWin, откройте метод TreeViewSelectionChanged() и измените его:

IF SELF:Owner:Owner:GetPaneClient( 1 ):AcCircuit:Pressed

IF ! IsInstanceOf(SELF:Owner:AcTcView:CurrentPage, #RaceSeason) 
// Новая строка
? "In Selection Method"

? "-------------------------"

ENDIF 








// Новая строка

ENDIF
Сохраните код и посмотрите на новую конструкцию IF. Функция IsInstanceOf() возвращает TRUE, если объект, переданный в первом параметре, является instance - представлением символьного имени, передаваемого во втором параметре. Первый параметр должен быть текущей страницей - <CurrentPage> элемента управления TreeView окна TabWin, однако, поскольку это защищенная переменная, то она в нашем классе она недоступна. Нам необходим специальный метод ACCESS, который мы назовем <AcTcView>, который еще требуется написать.

Закройте редактор, перейдите в модуль W:TabWin, создайте новый примитив и впишите в него:

ACCESS AcTcView( ) CLASS TabWin

RETURN SELF:oDCtcView
Это обеспечит видимость и доступность элемента управления Tab вне границ его класса и даст доступ к <CurrentPage>. Сохраните код, закройте редактор, скомпилируйте и запустите приложение.

После запуска приложения перейдите на вкладку Race Season и выберите какой-либо элемент в TreeView. Поскольку кнопка Circuit не нажата, мы не должны получить никаких сообщений. Теперь нажмите кнопку Circuit и выберите другой элемент во вкладке Race Season. Вновь не должно быть никаких сообщений. Чтобы проверить, что метод работает как нужно, проверьте его и в ситуации, когда он должен выдавать трассировочное сообщение.

Перейдите на вкладку View Detail и выберите любой элемент в TreeView Circuit. Если все идет по плану, на экране должно появиться трассировочное сообщение. Теперь мы готовы ввести данные в ListView в четвертой панели. По окончании тестирования закройте приложение и двинемся дальше.

Заполнение элемента ListView
В модуле W:TreeWin откройте метод TreeViewSelectionChanged() и измените его код следующим образом:

METHOD TreeViewSelectionChanged(oTreeViewSelectionEvent) CLASS TreeWin

LOCAL oItem AS TreeViewItem 




// Новая строка
LOCAL cKey AS STRING 





// Новая строка
SUPER:TreeViewSelectionChanged(oTreeViewSelectionEvent)

IF SELF:Owner:Owner:GetPaneClient(1):AcCircuit:Pressed

IF ! IsInstanceOf(SELF:Owner:AcTcView:CurrentPage, #RaceSeason)

// Начало новой секции ...

oItem := oTreeViewSelectionEvent:NewTreeViewItem 
// Строка 9

cKey := oItem:Value 




// Строка 10

SELF:Owner:Owner:GetPaneClient(4):Owner:LvCircuitFill(cKey)

// ... конец новой секции

ENDIF

ENDIF
RETURN NIL
Сохраните код и обратите внимание на две строки в новой секции метода (строки 9 и 10). В первой из них сохраняется вновь выбранный элемент, а во второй его значение присваивается переменной cKey.

Следующие строки хотя и выглядят запутанными, на самом деле весьма просты. Чтобы доказать это, давайте прочтен их слева направо. Мы уже рассмотрели цепочку от SELF до GetPaneClient(), в данный момент у нас имеется владелец и мы посылаем сообщение LvCircuitFill(), передавая одновременно в качестве параметра значение, содержащееся в переменной cKey. Для проверки метода сохраните код, закройте редактор, скомпилируйте и запустите приложение.

После запуска щелкните по кнопке Circuits, выберите элемент в TreeView, при этом должно появиться трассировочное сообщение, свидетельствующее о том, что мы находимся в ранее разработанном методе FillMethod(). По окончании тестирования закройте приложение и мы завершим ввод данных в ListView.

Перейдите в модуль W:SelectedRaces, откройте метод LVCircuitFill() и измените его: 

METHOD LVCircuitFill(cCircuit) CLASS SelectedRaces

LOCAL oItem AS ListViewItem
SELF:oDClvRaces:DeleteAll( )

SELF:Server:SetOrder('RacName')

SELF:Server:Seek(cCircuit)

DO WHILE SELF:Server:CirName == cCircuit .AND. ! SELF:Server:Eof

oItem := ListViewItem{ }

oItem:SetText(DToC(SELF:Server:RaceDate), #RaceDate)

oItem:SetText(SELF:Server:DriName, #DriName)

SELF:oDClvRaces:AddItem(oItem)

SELF:Server:Skip()

ENDDO
SELF:Server:SetOrder('RacDate')

RETURN NIL
Сохраните код и посмотрите на его первую строку. Там можно увидеть, что метод принимает строку в качестве параметра. Это ссылочный номер трассы, который мы передали как <cKey>. Мы начнем с того, что удалим все уже существующие элементы в ListView. Затем необходимо установить нужный ордер, после чего мы должны осуществить поиск (Seek) по полученному нами ключу.

Цикл DO WHILE будет исполняться до тех пор, пока этот ключ будет сохранять свое значение и мы не окажемся в конце файла. Внутри цикла создается элемент ListViewItem, сохраняем дату гонки (в литеральном представлении) в колонке #RaceDate, ссылочный номер пилота в колонке #DriName и добавляем этот элемент в общий список ListView. После этого указатель перемещается на следующую запись сервера. По окончании цикла восстанавливается ордер по датам и метод завершает свою работу.

Для проверки метода сохраните код, закройте редактор, скомпилируйте и запустите приложение.

Щелкните по кнопке Circuit и выберите одну из трасс в элементе TreeView. Вы должны увидеть соответствующие записи в окне ListView (рис. 21.15). В данный момент выведены лишь номера пилотов, что нельзя признать удачным решением. Закройте приложение и заменим номера фамилиями.

Фамилии - и никаких номеров

Вновь откройте метод LvCircuitFill() и измените его:

METHOD LVCircuitFill(cCircuit) CLASS SelectedRaces

LOCAL oItem AS ListViewItem

LOCAL oDrivers AS Drivers 




// Новая строка
LOCAL cKey AS STRING 





// Новая строка
oDrivers := Drivers{ } 





// Новая строка
SELF:oDClvRaces:DeleteAll()

SELF:Server:SetOrder('RacName')

SELF:Server:Seek(cCircuit)
DO WHILE SELF:Server:CirName == cCircuit .AND. ! SELF:Server:Eof

oItem := ListViewItem{ }

oItem:SetText(DToC(SELF:Server:RaceDate), #RaceDate)

cKey := SELF:Server:DriName 



// Новая строка
IF oDrivers:Seek(cKey) 




// Новая строка
oItem:SetText(oDrivers:DriName, #DriName) 
// Строка изменена
ENDIF 






// Новая строка

SELF:oDClvRaces:AddItem(oItem)


SELF:Server:Skip( )

ENDDO
SELF:Server:SetOrder('RacDate')

oDrivers:Close() 






// Новая строка
RETURN NIL
Добавление двух переменных и использование сервера Drivers не представляет из себя ничего нового. Однако, в цикл DO WHILE внесены важные изменения. Вместо того, чтобы помещать <DriName> непосредственно в создаваемый элемент ListViewItem, мы помещаем его во вновь созданную переменную <cKey> и используем это значения для поиска во вновь открытом сервере Drivers. Если поиск успешен, мы помещаем найденное значение <Drivers:DriName> в создаваемый элемент. В конце метода сервер закрывается. Сохраните код, закройте редактор, скомпилируйте и запустите приложение.

После запуска приложения щелкните по кнопку Circuits, выберите трассу в TreeView и в четвертой панели должны появиться фамилии пилотов (рис. 21.16). Это то, что мы хотели, и это работает, однако, не очень эффективно. Если этот метод должен быть запущен один раз, он вполне хорош, однако, в действительности он будет вызываться при каждом новом выборе элемента в элементе TreeView. Это означает, что сервер Drivers будет открываться и закрываться при каждом перемещении по ветвям дерева, что нельзя признать хорошей практикой и является весьма расточительной операций с точки зрения времени выполнения.

Для исправления ситуации у нас есть две возможности:

· мы можем отследить всю цепочку сообщений до окна F1SplitWindow и использовать сервер Drivers там, либо

· мы можем открыть сервер в окне SelectedRaces как защищенную переменную.

Мы используем второй вариант. По окончании тестирования закройте приложение.

Откройте окно SelectedRaces в оконном редакторе, перейдите на вкладку DataWindow в окне его свойств, щелкните по свойству Class Declaration и в редакторе кода измените объявление класса следующим образом:

CLASS SelectedRaces INHERIT DATAWINDOW

PROTECT oDClvRaces AS LISTVIEW

//{{%UC%}} USER CODE STARTS HERE (do NOT remove this line)

PROTECT oDrivers AS Drivers
Сохраните код, закройте редактор, и в нижней части вкладки откройте свойство PostInit. Добавьте в него следующую строку непосредственно перед возвратом из метода:

SELF:oDrivers := Drivers{}
Сохраните код, закройте редактор, перейдите на вкладку Window Events, щелкните свойство окна Close и измените существующий код:

METHOD Close(oEvent) CLASS SelectedRaces

SUPER:Close(oEvent)

//Put your changes here

SELF:oDrivers:Close()

RETURN NIL
Сохраните код, закройте редакторы, откройте метод LvCircuitFill() и удалите из него три следующие строки:

LOCAL oDrivers AS Drivers

oDrivers := Drivers{ }

oDrivers:Close( )
Теперь измените строку с IF и одну строку ниже таким образом:

IF SELF:oDrivers:Seek(cKey)

oItem:SetText(SELF:oDrivers:DriName, #DriName)
Все ссылки к <oDrivers> теперь должны иметь префикс SELF:, поскольку эта переменная является защищенной. Сохраните код, закройте редактор, скомпилируйте и запустите приложение. Если повторить предыдущий тест, мы не увидим никаких изменений на экране, однако нам известно, что внутренне код стал намного более эффективным. По окончании тестирования закройте приложение.

Управление разделенным окном SplitWindow

Четвертая панель требуется только при нажатии на кнопку Circuits, для двух других кнопок она не нужна. Занимаемое панелью место лучше использовать так, чтобы позволить второй панели занять его. Для того, чтобы добиться этого, перейдите в модуль C:F1SplitWindow, создайте новый примитив и в редакторе кода напишите:

METHOD SetPaneTwo( ) CLASS F1SplitWindow

LOCAL iHeight1 AS INT

LOCAL iHeight3 AS INT

LOCAL diSize AS dimension
iHeight1 := SELF:GetPaneSize(1):Height

iHeight3 := SELF:GetPaneSize(3):Height

diSize := SELF:GetPaneSize(2)

diSize:Height := iHeight1 + iHeight3

SELF:SetPaneSize(diSize, 2)

RETURN NIL

Сохраните код и давайте разберем его. В переменных <iHeight1> и <iHeight3> сохраняются размеры по высоте панелей 1 и 3, соответственно. Затем создается объект, представляющий размеры второй панели, а после этого высота второй панели устанавливается равной сумме высот первой и третьей панелей. В заключение размер второй панели устанавливается в соответствии с новым полученным значением высоты и метод завершает свою работу. Теперь давайте найдем место, куда вставить этот метод.

Откройте метод ButtonHandler() и измените его:

CASE symButton = #rbTeams

SELF:GetPaneClient(3):TeamsFill()

SELF:SetPaneTwo() 

// Новая строка

CASE symButton = #rbCircuits
Это выглядит хорошим местом для размещения обращения к новому методу. Он должен активизироваться при нажатии на кнопку Teams. Сохраните код, закройте редактор, скомпилируйте и запустите приложение. Оно должно выглядеть как обычно, а при нажатии на кнопку Teams вторая панель должна изменить свои размеры и занять всю правую часть экрана. Проблема теперь заключается в том, что при нажатии кнопки Circuit четвертая панель остается скрытой. Нам нужен способ для отмены изменения размера второй панели, поэтому закройте приложение и добавим необходимый код.

Прежде всего, подобное использование площади экрана вполне обоснованно, так что давайте сделаем его таковым по умолчанию. Перейдите в модуль C:Start, откройте метод Start() и дополните его одной строкой.

oSelectedRaces:Show()

oSplit:SetPaneTwo() 

// Новая строка

SELF:Exec()
Сохраните код и закройте редактор. Сейчас мы обеспечили открытие второй панели на всю правую часть экрана при запуске приложения, однако теперь при каждом нажатии на кнопку Teams приложение будет бесцельно перерисовывать эту же саму панель. Следовательно, нам нужен какой-то способ, чтобы определить, развернута ли вторая панель на всю ее высоту. Самым простым способом будет установить соответствующую переменную в  F1SplitWindow, что мы и сделаем.

Перейдите в модуль C:F1SplitWindow, откройте в редакторе кода объявление класса и добавьте в список защищенных переменных еще одну -

PROTECT oRaceYr AS RaceYr

PROTECT lPane2FullHeight AS LOGIC // Новая строка
Сохраните код, закройте редактор откройте метод SetPaneTwo() и измените его код:

METHOD SetPaneTwo( ) CLASS F1SplitWindow

LOCAL iHeight1 AS INT

LOCAL iHeight3 AS INT

LOCAL diSize AS dimension
IF ! SELF:lPane2FullHeight 


// Новая строка
iHeight1 := SELF:GetPaneSize( 1 ):Height

iHeight3 := SELF:GetPaneSize( 3 ):Height

diSize := SELF:GetPaneSize( 2 )

diSize:Height := iHeight1 + iHeight3

SELF:SetPaneSize(diSize, 2)

SELF:lPane2FullHeight := TRUE 

// Новая строка
ENDIF 





// Новая строка

RETURN NIL
Первая из новых строк проверит, действительно ли вторая панель максимально развернута и предупредит бессмысленную ее перерисовку. Вторая из добавленных строк присвоит переменной значение TRUE, если пришлось изменить размер панели с маленького на большой. Однако, до сих пор мы не обеспечили возврат на экран четвертой панели, давайте сделаем это. Перейдите в конец метода SetPaneTwo(), пропустите одну строку и добавьте следующий код:

METHOD SetPaneFour() CLASS F1SplitWindow

LOCAL iHeight2 AS INT

LOCAL diSize AS dimension
IF SELF:lPane2FullHeight

iHeight2 := INT(SELF:GetPaneSize(2):Height / 2)

diSize := SELF:GetPaneSize(2)

diSize:Height := iHeight2

SELF:SetPaneSize(diSize, 2)

SELF:lPane2FullHeight := FALSE

ENDIF
RETURN NIL
Этот код просто проверит высоту второй панели, и, если она максимальна, просто разделит имеющуюся высоту пополам и присвоит новые размеры панели. Однако, результат деления может оказаться дробным, а мы заинтересованы в целочисленных значениях, поэтому, прежде, чем  присвоить значение высоты панели, мы округляем значение с помощью функции INT. Сохраните код, закройте редактор, откройте метод ButtonHandler() и измените его:

METHOD ButtonHandler(symButton) CLASS F1SplitWindow
DO CASE

CASE symButton = #rbDrivers

SELF:GetPaneClient(3):DriversFill()

SELF:SetPaneTwo() 

// Новая строка

CASE symButton = #rbTeams

SELF:GetPaneClient(3):TeamsFill()

SELF:SetPaneTwo()

CASE symButton = #rbCircuits

SELF:GetPaneClient(3):CircuitFill()

SELF:SetPaneFour() 

// Новая строка

ENDCASE
RETURN NIL
Первая из добавленных строк развернет вторую панель полностью при нажатии на кнопку Drivers (эта же кнопка используется и по умолчанию). Вторая новая строка разделит вторую панель пополам при нажатии на кнопку Circuits. Для проверки нового кода сохраните его, закройте редактор, скомпилируйте и запустите приложение.

Теперь (рис. 21.17) список гонок развернут на всю правую половину окна по умолчанию. При нажатии на кнопку Circuits список изменяет свою высоту. По окончании тестирования закройте приложение.

Добавление значков в элемент ListView
При больших списках зачастую трудно отличить одну запись от другой. Использование значков в элементах TreeView и ListView зачастую может помочь в этом, тем более, что их добавление реализуется очень просто. В 20-й главе мы использовали редактор изображений для создания значок, сейчас мы повторно используем их.

Создайте новый модуль "I:Icons", откройте редактор кода и импортируйте файл Ico.prg из каталога C:\EVO\F1. Удалите текстовый блок, сохраните код и закройте редактор. Импортированный сейчас код тот же, что был генерирован ранее, при создании значков. Если вы используете другие имена дисков и путей, внесите необходимые коррективы в код.

Перейдите в модуль W:SelectedRaces, откройте метод PostInit() и добавьте в него следующие строки:

LOCAL oColumn AS ListViewColumn

LOCAL oImageList AS ImageList 


// Новая строка
oImageList := ImageList{ 7, Dimension{ 16, 16 } } // Новая строка
oImageList:Add(FirstIco{}) 


// Новая строка
oImageList:Add(SecondIco{}) 


// Новая строка
oImageList:Add(ThirdIco{}) 


// Новая строка
oImageList:Add(FourthIco{}) 


// Новая строка
oImageList:Add(FifthIco{}) 


// Новая строка
oImageList:Add(SixthIco{}) 


// Новая строка
oImageList:Add(NFIco{}) 


// Новая строка
SELF:oDClvRaces:SmallImageList := oImageList 
// Новая строка
SELF:oDClvRaces:LargeImageList := oImageList 
// Новая строка
oHyperLabel := HyperLabel{ #RaceDate, 'Date'}
Сохраните код и посмотрите, что в нем происходит. Ранее были созданы семь значков, размерами 16х16 пикселей. Объект ImageList включает в себя семь элементов, каждый размеров 16х16. Далее создаются значки и добавляются в объект по порядку, поскольку позднее мы будет обращаться к ним по номерам элементов. Если ваша машина настроена на использование больших значков, они будут автоматически увеличены до размеров 32х32 и будут выглядеть весьма странно.

Закройте редактор, откройте метод LVCircuit() и добавьте в него несколько строк:

IF SELF:oDrivers:Seek(cKey)

oItem:SetText(SELF:oDrivers:DriName, #DriName)

ENDIF
IF SELF:Server:FiniPos > 0 .AND. SELF:Server:FiniPos < 7 

// Новая строка
oItem:ImageIndex := SELF:Server:FiniPos 

// Новая строка
ELSE 







// Новая строка

oItem:ImageIndex := 7 




// Новая строка

ENDIF 







// Новая строка

SELF:oDClvRaces:AddItem(oItem)
Сохраните код и изучите его. У нас есть шесть значков для представления первых шести мест и один значок для гонщиков, занявших другие места. Если финишная позиция (FiniPos), указанная в сервере, менее 7, то нужно вывести соответствующий значок. Поскольку значки добавлялись в список значок по порядку, то занятое место и номер значка соответствуют друг другу. Обратите внимание, что коль скоро один из элементов имеет свой значок, значит и остальные также должны иметь какой-то значок. Значком по умолчанию является первый значок в списке.

Сохраните код, закройте редактор, скомпилируйте и запустите приложение. При появлении его на экране нажмите на кнопку Circuits и выберите один из элементов в дереве TreeView, чтобы в четвертой панели отобразились соответствующие данные. Если все работает нормально, напротив каждого из элементов в четвертой панели будет выведен соответствующий значок.

Изменение формы курсора мыши

Последнее, что мы намерены сделать, это изменить форму указателя мыши, создав курсор F1Car. Однако, необходимое предупреждение. Из трех компьютеров, на которых испытывались приводимые здесь примеры, один падал всякий раз, как только курсор F1Car добавлялся в приложение. Приложение работало без сбоев на двух компьютерах (Windows95/200/128 и WindowsNT/350/160), однако неизменно приводило к краху систему с Windows98(550/192). Ко времени печати настоящей книги не удалось установить причину краха - то ли виновата ОС, то ли процессор, то ли что-то иное…

Указатель F1Car может быть добавлен во все приложение, либо в какие-то отдельные панели. Давайте попробуем добавить его в панель 1. Откройте метод App:Start() и измените секцию <oShowWin> следующим образом:

oShowWin := ShowWin{oSplit}

oSplit:SetPaneSize(oShowWin:Size, 1)

oShowWin:Pointer := F1CAR{} 

// Новая строка
oShowWin:Show( )

oSplit:SetPaneClient(oShowWin, 1)
Сохраните код, закройте редактор, скомпилируйте и запустите приложение.

Захватите вертикальный разделитель мышью и растяните первую панель, пока она не примет вид, показанный на рис. 21.19. Если теперь поместить указатель на первую панель, правее радиокнопок, он должен приять вид, показанный на экране. Передвиньте указатель в область кнопок или выведите его из панели, он должен принять нормальный вид.

 А теперь послушайте…

На этом заканчивается описание элементов TreeView и ListView в настоящей главе. Однако, прежде, чем мы перейдем к заключению, давайте немного повеселимся. Те, у кого имеются звуковые карты в компьютерах, могут захотеть использовать эту возможность в своих приложениях. CA-Visual Objects позволяет сделать это с легкостью необыкновенной. Для короткой демонстрации перейдите в модуль C:Start, откройте метод Start() и добавьте одну строку ближе к началу метода -

LOCAL oSelectedRaces AS SelectedRaces

sndPlaySound(PSZ("C:\EVO\F1\Start.Wav"), 0) 
// Новая строка
SetDefault("C:\Evo\F1")
а ближе к концу метода добавьте следующее:

SELF:Exec( )

sndPlaySound(PSZ("C:\EVO\F1\End.Wav"), 0) 
// Новая строка
RETURN NIL
Убедитесь в правильности указанного пути, сохраните код, закройте редактор, скомпилируйте и запустите приложение.

Как вы можете видеть, приложение ожидает окончания работы функции SndPlaySound(), прежде чем начнется создание окна и сама работа. Поскольку длительность воспроизведения составляет около 16 сек., а многие люди хотят, чтобы их программы работали как можно быстрее, такая задержка не очень практичный подход.

Однако, это показывает, как легко воспроизводятся звуковые файлы и в главе 23 мы покажем, как приложение может работать одновременно с воспроизведением музыки, однако на сей момент довольно. Закройте приложение и послушайте музыку при закрытии приложения.

Заключение

В этой главе мы создали приложение с самого начала.

Используя функции семейства DB…, мы создали файлы данных и индексов.

Мы создали серверы и окна и встроили их в разделенное окно SplitWindow.

Мы представили такие элементы управления как TabContol, TreeView и ListView, а последний - в обычной версии и версии DataListView и заставили их реагировать на выбор кнопки RadioButton.

Мы создали панели в окне SplitWindow и обеспечили изменение их размеров.

Мы добавили значки в элемент ListView и курсор пользователя в одну из панелей.

И, в конце, мы добавили звук в наше приложение.

И, хотя приложение далеко от идеала, можно с легкостью создать дополнительные окна для управления приложением.

В нашем приложении мы использовали класс SplitWindow для работы с несколькими окнами, однако, если вы создаете приложение, использующее стиль Проводника, в CA-Visual Objects уже имеется класс ExplorerWindow.

